

# SCIENCE & VIE

MENSUEL N° 880

JANVIER 1991

**PLAN  
DÉTAILLÉ  
POUR FAIRE  
UN HOMME**

145 FB - 6 FS - \$can 3,75 - 550 Ptas - 550 Esc - 28 Dh - 2.750 Dt - 4800 L - USA NYC \$ 4.25 - RCI : 1370 CFA - 940 CFP

● **La première  
vraie carte  
du monde**

M 2578 - 880 - 20,00 F



**DROGUE:  
LE PLAISIR QUI TUE (2)**

Jégo.

# Le Rubik's cube

INFORMATIQUE AMUSANTE

**A**u début des années 80, venu tout droit de Hongrie et inventé par Enno Rubik, arrivait en France le Rubik's cube. Ce casse-tête, d'ailleurs plus difficile qu'hongrois, se composait d'un cube dont chaque face comportait 9 carrés de base. Par rotation ou translation de groupes d'entre eux il était simple et rapide de détruire la belle organisation de chaque face mais... beaucoup plus délicat de revenir à la situation de départ. Ce mois-ci, nous vous proposerons donc d'adapter le Rubik's cube à l'informatique. Notre programme remplacera le cube en affichant chacune de ses faces sur l'écran et permettra, par l'intermédiaire du clavier, de déplacer les carrés de base. De plus, nous avons muni ce programme d'une option complémentaire que nous avons baptisée PLAY BACK. Après qu'une série de déplacements a été effectuée, si cette option est demandée, tous vos coups seront visualisés les uns après les autres, en marche arrière, jusqu'au retour à la disposition de départ du cube, soit chaque face d'une couleur unique. De même, la fonction CONTINUER ? restituera l'état du cube après le dernier coup joué afin de pouvoir poursuivre le jeu. Ces quelques précisions apportées, passons à l'écriture du programme.

En premier lieu nous demanderons à l'Amstrad de fonctionner en mode 0, ceci afin de disposer d'une palette de couleurs suffisante pour différencier chacune des faces du cube ; puis nous redéfinirons les symboles graphiques nécessaires à la représentation des carrés de base. Ensuite nous créerons une série de tableaux de mémorisation à l'aide de l'instruction DIM ; A, B, C, D, E et F auront pour mission de mémoriser l'état de chaque face avant et après un coup. De même T\$ et S\$ mémoriseront respectivement "numéro de tranche" et "direction" de chaque coup ; enfin P mémorisera trois carrés adjacents de manière à pouvoir effectuer, sur l'écran, la translation ou la rotation adéquate.

Vient ensuite la page de présen-

tation. Nous initialiserons nos tableaux de mémorisation de manière à présenter les faces du cube en position de départ. Son affichage, chaque face déployée, sera assuré par une première sous-routine et une vue en perspective ne prenant pas en compte les couleurs, et prendra place en haut et à droite de l'écran. Sur cette dernière sera indiqué le numéro de chaque tranche (de 1 à 6) ceci afin de remémorer au joueur leur disposition tout au long de la partie.

Ensuite vient le début du jeu. La machine demandera qu'on lui indique sur quelle tranche doit être effectué le mouvement, puis fera préciser sa direction ; gauche ou droite pour les tranches 1 à 3, haut ou bas pour celles de 4 à 6. A chaque fois l'ordinateur contrôlera la cohérence du coup ; il est en effet impossible de déplacer, par exemple, la tranche 1 vers le haut. Ces indications fournies, le programme s'aiguillera vers une série de sous-routines de déplacement. Les deux premières concernent les translations et donc les mouvements effectués sur les tranches 2 et 5 ; la troisième concerne les rotations des faces supérieures et inférieures, donc les tranches 1 et 3 et enfin, la dernière, les faces latérales du cube, donc les tranches 4 et 6.

- La sous-routine de translation : ici deux cas sont à distinguer, la translation à droite ou la translation à gauche. Notre sous-routine comportera donc deux parties fonctionnant sur le même principe mais dont l'ordre de permutation sera inverse. Dans les deux cas, P mémorisera le dernier triplet sur lequel portera le coup puis les déplacements seront effectués vers celui-ci et les valeurs contenues dans P seront utilisées pour modifier le premier triplet. Nous pratiquerons de même pour les translations verticales.

- Les sous-routines de rotation de face : ici encore P prendra la valeur du dernier triplet concerné et nous effectuerons des permutations circulaires sur le tableau de mémorisation de la face concernée. Précisons ici que nous aurons

toutefois recours en parallèle à la sous-routine de translation ; ceci afin de conserver la cohérence du cube. Un élément placé sur un bord, par exemple sera toujours constitué des deux mêmes couleurs. Lors de son déplacement sur une face il faudra donc le répercuter sur la face contiguë.

• La sous-routine concernant le mode PLAY BACK : nous utilise-

rons ici les tableaux T\$ et S\$. En effet, ces derniers ayant mémorisé les mouvements de chaque coup en les lisant à l'envers et en inversant les directions, nous serons en mesure de revenir à la situation de départ en faisant défiler la partie en marche arrière.

L'instruction MODE détermine le nombre de caractères par ligne d'écran, et influe sur la palette de

couleurs. SYMBOL AFTER libère une zone de la grille de caractères standards pour pouvoir les modifier. A cette instruction sera associé SYMBOLE, qui sera utilisée pour définir chaque nouveau graphique. Le premier nombre indiquera le numéro de symbole à utiliser pour l'appeler par CHR\$(X), les 8 suivants la position des points. BORDER, PAPER et PEN

```

10 MODE 0:BORDER 1:PAPER 0:CLS
20 CLEAR
30 LET COUP=1
40 REM *****
50 REM *
60 REM * REDEFINITION DES SYMBOLES *
70 REM * GRAPHIQUES. *
80 REM *
90 REM *****
100 SYMBOL AFTER 127
110 SYMBOL 128,0,124,254,254,254,254,254,254,254
120 SYMBOL 129,254,254,254,254,254,254,124,0
130 REM *****
140 REM *
150 REM * CREATION DES TABLEAUX DE *
160 REM * MEMO. DE L'ETAT DES FACES. *
170 REM *
180 REM *****
190 DIM A(3,3)
200 DIM B(3,3)
210 DIM C(3,3)
220 DIM D(3,3)
230 DIM E(3,3)
240 DIM F(3,3)
250 DIM P(3)
260 REM *****
270 REM *
280 REM * CREATION DES TABLEAUX DE *
290 REM * MEMO. DES COUPS DU JOUEUR. *
300 REM *
310 REM *****
320 DIM T$(300)
330 DIM S$(300)
340 REM *****
350 REM *
360 REM * INITIALISATION DES TABLEAUX. *
370 REM *
380 REM *****
390 FOR X=1 TO 3
400 FOR Y=1 TO 3
410 LET A(X,Y)=1
420 LET B(X,Y)=3
430 LET C(X,Y)=6
440 LET D(X,Y)=7
450 LET E(X,Y)=12
460 LET F(X,Y)=4
470 NEXT Y
480 NEXT X
490 REM *****
500 REM *
510 REM * AFFICHAGE DE L'ECRAN DE JEU. *
520 REM *
530 REM *****
540 GOSUB 1250
550 PEN 1
560 PLOT 450,350,12
570 DRAW 570,350
580 DRAW 570,230
590 DRAW 450,230
600 DRAW 450,350
610 DRAW 500,390
620 DRAW 610,390
630 DRAW 610,280
640 DRAW 570,230
650 PLOT 570,350
660 DRAW 610,390
670 FOR I=1 TO 2
680 PLOT 450+(40*I),230
690 DRAW 450+(40*I),350
700 DRAW 500+(37*I),390
710 PLOT 450,230+(40*I)
720 DRAW 570,230+(40*I)
730 DRAW 610,280+(37*I)
740 PLOT 450+(20*I),350+(14*I)
750 DRAW 570+(14*I),350+(14*I)
760 DRAW 570+(14*I),230+(20*I)
770 NEXT I
780 FOR I=1 TO 3
790 LOCATE 12,(3*I)+2:PRINT I
800 NEXT I
810 LOCATE 15,13:PRINT "4 5 6"
820 LOCATE 17,19:PRINT CHR$(240):LOCATE 17,21:PRINT "H"
830 LOCATE 14,22:PRINT CHR$(242);" G+D ";CHR$(243)
840 LOCATE 17,23:PRINT "B":LOCATE 17,25:PRINT CHR$(241)
850 REM *****
860 REM *
870 REM * CHOIX DU COUP A JOUER. *
880 REM *
890 REM *****
900 PEN 1:LOCATE 1,1:PRINT "TRANCHE? "
910 LET T$=INKEY$:IF T$="P" OR T$="p" THEN GOSUB 2
920 GOTO 900
930 IF T$<>"1" AND T$<>"2" AND T$<>"3" AND T$<>"4" AND T$<>"5" AND T$<>"6" THEN GOTO 910
940 LOCATE 1,1:PRINT "DIRECTION?"
950 LOCATE 1,3
960 IF T$="1" OR T$="2" OR T$="3" THEN PRINT "(G O U D)":LET D=0
970 IF T$="4" OR T$="5" OR T$="6" THEN PRINT "(H O U B)":LET D=1
980 LET D$=UPPER$(INKEY$):IF D$="" THEN GOTO 970
990 IF D=0 AND D$<>"G" AND D$<>"D" THEN GOTO 970
1000 IF D=1 AND D$<>"H" AND D$<>"B" THEN GOTO 970
1000 LOCATE 1,3:PRINT " "
1010 LOCATE 1,1:PRINT " "
1020 LET T$(COUP)=T$
1030 LET S$(COUP)=D$
1040 LET COUP=COUP+1
1050 REM *****
1060 REM *
1070 REM * AIGUILLAGE DU PROGRAMME VERS *
1080 REM * LA SOUS ROUTINE CONCERNEE. *
1090 REM *
1100 REM *****
1110 IF T$="2" THEN LET TR=2:GOSUB 1560
1120 IF T$="5" THEN LET TR=2:GOSUB 1790
1130 IF T$="1" THEN LET TR=1:GOSUB 2020
1140 IF T$="3" THEN LET TR=3:GOSUB 2130
1150 IF T$="4" THEN LET TR=1:GOSUB 2240
1160 IF T$="6" THEN LET TR=3:GOSUB 2350
1170 GOSUB 1250
1180 GOTO 900
1190 REM *****
1200 REM *
1210 REM * SOUS ROUTINE D'AFFICHAGE *
1220 REM * DU CUBE. *
1230 REM *
1240 REM *****
1250 PAPER 5
1260 FOR H=1 TO 3
1270 FOR V=1 TO 3
1280 PEN A(H,V)
1290 LOCATE H+6,(2*V)+6:PRINT CHR$(128)
1300 LOCATE H+6,(2*V)+7:PRINT CHR$(129)
1310 PEN B(H,V)
1320 LOCATE H,(2*V)+12:PRINT CHR$(128)
1330 LOCATE H,(2*V)+13:PRINT CHR$(129)
1340 PEN C(H,V)
1350 LOCATE H+3,(2*V)+12:PRINT CHR$(128)
1360 LOCATE H+3,(2*V)+13:PRINT CHR$(129)
1370 PEN D(H,V)
1380 LOCATE H+6,(2*V)+12:PRINT CHR$(128)
1390 LOCATE H+6,(2*V)+13:PRINT CHR$(129)
1400 PEN E(H,V)

```

fixent respectivement la couleur des bords de l'écran, du fond et des caractères. LOCATE X, Y adresse le curseur en ligne-colonne sur l'écran. Ses équivalents les plus courants sont HTAB-VTAB, PRINT AT. Les valeurs prises par X varient de 1 à 21 et celles de Y, de 1 à 25.

L'utilisation de ce programme est très simple. Après l'avoir entièrement frappé et demandé RUN, le

cube s'affichera sur l'écran. Puis TRANCHE ? sera demandé. Il suffira, en s'aidant de la représentation du cube en perspective, de taper le numéro de la tranche sur laquelle on souhaite agir. Dès lors, DIRECTION ?, suivi de "(D ou G)" ou de "(H ou B)" en fonction de la tranche choisie, sera demandé. Une fois la réponse fournie, la représentation du cube sera modifiée conformé-

ment à la demande de mouvement. Pour accéder au mode PLAY BACK il suffira de taper "P" au lieu d'un numéro de tranche. La partie défilera en marche arrière jusqu'à la situation de départ. Là on répondra à la question CONTINUER ? par oui ou par non selon que l'on désire reprendre la partie au dernier coup joué ou, inversement, repartir sur de nouvelles bases.

**Henri-Pierre Penel**

```

1410 LOCATE H+9, (2*V)+12:PRINT CHR$(128)
1420 LOCATE H+9, (2*V)+13:PRINT CHR$(129)
1430 PEN F(H,V)
1440 LOCATE H+6, (2*V)+18:PRINT CHR$(128)
1450 LOCATE H+6, (2*V)+19:PRINT CHR$(129)
1460 NEXT V
1470 NEXT H
1480 PAPER 0
1490 RETURN
1500 REM *****
1510 REM *
1520 REM * SOUS ROUTINE DE TRANSLATION *
1530 REM * HORIZONTALE DES CARRES. *
1540 REM *
1550 REM *****
1560 IF D$="D" THEN GOTO 1640
1570 FOR I=1 TO 3
1580 LET P(I)=C(I,TR)
1590 LET C(I,TR)=D(I,TR)
1600 LET D(I,TR)=E(I,TR)
1610 LET E(I,TR)=B(I,TR)
1620 LET B(I,TR)=P(I)
1630 NEXT I
1640 IF D$="G" THEN GOTO 1720
1650 FOR I=1 TO 3
1660 LET P(I)=C(I,TR)
1670 LET C(I,TR)=B(I,TR)
1680 LET B(I,TR)=E(I,TR)
1690 LET E(I,TR)=D(I,TR)
1700 LET D(I,TR)=P(I)
1710 NEXT I
1720 RETURN
1730 REM *****
1740 REM *
1750 REM * SOUS ROUTINE DE TRANSLATION *
1760 REM * VERTICALE DES CARRES. *
1770 REM *
1780 REM *****
1790 IF D$="B" THEN GOTO 1870
1800 FOR I=1 TO 3
1810 LET P(I)=D(TR,I)
1820 LET D(TR,I)=F(TR,I)
1830 LET F(TR,I)=B(4-TR,I)
1840 LET B(4-TR,I)=A(TR,I)
1850 LET A(TR,I)=P(I)
1860 NEXT I
1870 IF D$="H" THEN GOTO 1950
1880 FOR I=1 TO 3
1890 LET P(I)=D(TR,I)
1900 LET D(TR,I)=A(TR,I)
1910 LET A(TR,I)=B(4-TR,I)
1920 LET B(4-TR,I)=F(TR,I)
1930 LET F(TR,I)=P(I)
1940 NEXT I
1950 RETURN
1960 REM *****
1970 REM *
1980 REM * SOUS ROUTINES DE ROTATION *
1990 REM * DES FACES. *
2000 REM *
2010 REM *****
2020 GOSUB 1560
2030 FOR I=1 TO 3
2040 LET P(I)=A(I,3)
2050 NEXT I
2060 IF D$="D" THEN GOTO 2090
2070 LET A(2,3)=A(3,2):LET A(3,2)=A(2,1):LET A(2,1)
)=A(1,2):LET A(1,2)=P(2)
2080 LET A(1,3)=A(3,3):LET A(3,3)=A(3,1):LET A(3,1)
)=A(1,1):LET A(1,1)=P(1)
2090 IF D$="G" THEN GOTO 2120
2100 LET A(2,3)=A(1,2):LET A(1,2)=A(2,1):LET A(2,1)
)=A(3,2):LET A(3,2)=P(2)
2110 LET A(1,3)=A(1,1):LET A(1,1)=A(3,1):LET A(3,1)
)=A(3,3):LET A(3,3)=P(1)
2120 RETURN
2130 GOSUB 1560
2140 FOR I=1 TO 3
2150 LET P(I)=F(I,3)
2160 NEXT I
2170 IF D$="D" THEN GOTO 2200
2180 LET F(2,3)=F(3,2):LET F(3,2)=F(2,1):LET F(2,1)
)=F(1,2):LET F(1,2)=P(2)
2190 LET F(1,3)=F(3,3):LET F(3,3)=F(3,1):LET F(3,1)
)=F(1,1):LET F(1,1)=P(1)
2200 IF D$="G" THEN GOTO 2230
2210 LET F(2,3)=F(1,2):LET F(1,2)=F(2,1):LET F(2,1)
)=F(3,2):LET F(3,2)=P(2)
2220 LET F(1,3)=F(1,1):LET F(1,1)=F(3,1):LET F(3,1)
)=F(3,3):LET F(3,3)=P(1)
2230 RETURN
2240 GOSUB 1790
2250 FOR I=1 TO 3
2260 LET P(I)=C(I,3)
2270 NEXT I
2280 IF D$="H" THEN GOTO 2310
2290 LET C(2,3)=C(3,2):LET C(3,2)=C(2,1):LET C(2,1)
)=C(1,2):LET C(1,2)=P(2)
2300 LET C(1,3)=C(3,3):LET C(3,3)=C(3,1):LET C(3,1)
)=C(1,1):LET C(1,1)=P(1)
2310 IF D$="B" THEN GOTO 2340
2320 LET C(2,3)=C(1,2):LET C(1,2)=C(2,1):LET C(2,1)
)=C(3,2):LET C(3,2)=P(2)
2330 LET C(1,3)=C(1,1):LET C(1,1)=C(3,1):LET C(3,1)
)=C(3,3):LET C(3,3)=P(1)
2340 RETURN
2350 GOSUB 1790
2360 FOR I=1 TO 3
2370 LET P(I)=E(I,3)
2380 NEXT I
2390 IF D$="H" THEN GOTO 2420
2400 LET E(2,3)=E(3,2):LET E(3,2)=E(2,1):LET E(2,1)
)=E(1,2):LET E(1,2)=P(2)
2410 LET E(1,3)=E(3,3):LET E(3,3)=E(3,1):LET E(3,1)
)=E(1,1):LET E(1,1)=P(1)
2420 IF D$="B" THEN GOTO 2450
2430 LET E(2,3)=E(1,2):LET E(1,2)=E(2,1):LET E(2,1)
)=E(3,2):LET E(3,2)=P(2)
2440 LET E(1,3)=E(1,1):LET E(1,1)=E(3,1):LET E(3,1)
)=E(3,3):LET E(3,3)=P(1)
2450 RETURN
2460 REM *****
2470 REM *
2480 REM * SOUS ROUTINE "PLAY BACK". *
2490 REM *
2500 REM *****
2510 LET PBACK=COUP
2520 FOR B=PBACK TO 2 STEP -1
2530 LET PBACK=PBACK-1
2540 LET T$=T$(PBACK)
2550 IF S$(PBACK)="G" THEN LET D$="D"
2560 IF S$(PBACK)="D" THEN LET D$="G"
2570 IF S$(PBACK)="H" THEN LET D$="B"
2580 IF S$(PBACK)="B" THEN LET D$="H"
2590 IF T$="2" THEN LET TR=2:GOSUB 1560
2600 IF T$="5" THEN LET TR=2:GOSUB 1790
2610 IF T$="1" THEN LET TR=1:GOSUB 2020
2620 IF T$="3" THEN LET TR=3:GOSUB 2130
2630 IF T$="4" THEN LET TR=1:GOSUB 2240
2640 IF T$="6" THEN LET TR=3:GOSUB 2350
2650 LOCATE 1,1:PRINT "COUP No.":PBACK
2660 GOSUB 1250
2670 LOCATE 1,1:PRINT CHR$(7):FOR T=1 TO 2000:NEXT
T
2680 NEXT B
2690 LET D$="":LET T$="":LET COUP=1
2700 RETURN

```